

Example of Cross-Classified Random Effects Models: Students Changing Classrooms over Time Outcome: Teacher-Perceived Academic Effort (*see chapter 11b example materials online*)

This example is from chapter 11b of Hoffman (2015), in which 486 students (54% boys) were in 58 different classes at each of three annual occasions (beginning in grades 3, 4, or 5 in year 0). Comparisons with a saturated means, unstructured variance model for our outcome of time-varying perceived academic effort (*teff*) across years within students indicated piecewise linear change and a random intercept only could approximate the means, variances, and covariances across students. We continue by examining whether the effect of being in a given class operates while in the class (a year-specific, acute effect) or stays with each student after being in the class (a cumulative, transfer effect), followed by effects of predictors at each level (grade, gender, and time-varying teacher-perceived student aggression).

SAS syntax to set up the data:

We first need to transpose our data to create year-specific class ID variables from year-varying class ID.

```
* Transposing classroom to multivariate and merging back in;
PROC SORT DATA=classroom; BY ID year; RUN;
DATA teacher; SET classroom; KEEP class StudentID year; RUN;
PROC TRANSPOSE DATA=teacher OUT=teacherT PREFIX=class; BY StudentID; ID year; VAR class; RUN;
DATA classroom; MERGE classroom teacherT; BY StudentID; DROP _NAME_ _LABEL_; RUN;
```

The second step is to define two sets of dummy codes for whether a class has had an effect yet—these will serve as custom fixed or random intercepts in our models.

```
DATA classroom; SET classroom;
* Fixing missing values;
  IF class0=. THEN class0=-99;
  IF class1=. THEN class1=-99;
  IF class2=. THEN class2=-99;
* Dummy codes for acute (non-transfer) effects of classrooms across time;
  IF year=0 THEN DO; aclass0=1; aclass1=0; aclass2=0; END;
ELSE IF year=1 THEN DO; aclass0=0; aclass1=1; aclass2=0; END;
ELSE IF year=2 THEN DO; aclass0=0; aclass1=0; aclass2=1; END;
* Dummy codes for cumulative (transfer) effects of classrooms across time;
  IF year=0 THEN DO; tclass0=1; tclass1=0; tclass2=0; END;
ELSE IF year=1 THEN DO; tclass0=1; tclass1=1; tclass2=0; END;
ELSE IF year=2 THEN DO; tclass0=1; tclass1=1; tclass2=1; END;
* Setting missing classroom values to have no effect;
  IF class0=-99 THEN DO; tclass0=0; aclass0=0; END;
  IF class1=-99 THEN DO; tclass1=0; aclass1=0; END;
  IF class2=-99 THEN DO; tclass2=0; aclass2=0; END;
RUN;
```

Because the class0-2 variables will be used in a CLASS statement, any student missing any occasion will be deleted if any of these are missing. To prevent this, we change missing class ID values to -99.

Here we shut off the intercepts for the class IDs = -99 with missing values.

Here is what the data now look like for two students:

Year-specific class effects

Transfer class effects

Student	Class	Grade	Year	Class0	Class1	Class2	Aclass0	Aclass1	Aclass2	Tclass0	Tclass1	Tclass2
101	1	3	0	1	-99	43	1	0	0	1	0	0
101	-99	4	1	1	-99	43	0	0	0	0	0	0
101	43	5	2	1	-99	43	0	0	1	1	0	1
102	3	3	0	3	21	42	1	0	0	1	0	0
102	21	4	1	3	21	42	0	1	0	1	1	0
102	42	5	2	3	21	42	0	0	1	1	1	1

Model 1: Ignoring Effects of Class (2-level model; piecewise means, student random intercept)

$$\text{Effort}_{\text{tsc}} = \gamma_{000} + \gamma_{100}(\text{Year01}_{\text{tsc}}) + \gamma_{200}(\text{Year12}_{\text{tsc}}) + U_{0s0} + e_{\text{tsc}}$$

```
TITLE1 "Piecewise Means, Student Random Intercept Ignoring Class Effects";
PROC MIXED DATA=classroom COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS StudentID grade year class0 class1 class2;
  MODEL teff = year01 year12 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=StudentID;
  REPEATED year / TYPE=VC SUBJECT=StudentID;
  ODS OUTPUT InfoCrit=FitPInt; RUN;
```

```
Dimensions
Covariance Parameters          2
Columns in X                   3
Columns in Z Per Subject       1
Subjects                       486 → number of students
Max Obs Per Subject            3
```

```
Covariance Parameter Estimates
Cov Parm  Subject  Estimate  Standard Error  Z Value  Pr > Z
UN(1,1)   StudentID  0.5821   0.04968         11.72   <.0001  Random intercept variance across students
year      StudentID  0.3979   0.02091         19.03   <.0001  Residual variance within students
```

```
Null Model Likelihood Ratio Test
DF  Chi-Square  Pr > ChiSq
1   347.42     <.0001
```

The student random intercept variance is significantly > 0 (ICC = .589).

```
Information Criteria
Neg2LogLike  Parms  AIC  AICC  HQIC  BIC  CAIC
3067.7       2     3071.7  3071.7  3075.0  3080.1  3082.1
```

```
Solution for Fixed Effects
Effect      Estimate  Standard Error  DF  t Value  Pr > |t|
Intercept   3.9056   0.04765         872  81.97   <.0001
year01      -0.1706  0.04558         764  -3.74   0.0002
year12      0.01833  0.04605         770  0.40    0.6907
```

Model 2: Level-1 Fixed Effects for Year-Specific (Acute) Effects of Class (from Equation 11.13)

$$\begin{aligned} \text{Effort}_{\text{tsc}} = & \gamma_{000} + \gamma_{100}(\text{Year01}_{\text{tsc}}) + \gamma_{200}(\text{Year12}_{\text{tsc}}) + U_{0s0} + e_{\text{tsc}} \\ & + \gamma_{001}^0(\text{Class1}_c)(\text{Int0}_{\text{tsc}}) + \gamma_{002}^0(\text{Class2}_c)(\text{Int0}_{\text{tsc}}) \cdots + \gamma_{00c}^0(\text{ClassC}_c)(\text{Int0}_{\text{tsc}}) \\ & + \gamma_{001}^1(\text{Class1}_c)(\text{Int1}_{\text{tsc}}) + \gamma_{002}^1(\text{Class2}_c)(\text{Int1}_{\text{tsc}}) \cdots + \gamma_{00c}^1(\text{ClassC}_c)(\text{Int1}_{\text{tsc}}) \\ & + \gamma_{001}^2(\text{Class1}_c)(\text{Int2}_{\text{tsc}}) + \gamma_{002}^2(\text{Class2}_c)(\text{Int2}_{\text{tsc}}) \cdots + \gamma_{00c}^2(\text{ClassC}_c)(\text{Int2}_{\text{tsc}}) \end{aligned}$$

```
TITLE1 "Fixed effects of class -- acute";
PROC MIXED DATA=classroom COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS StudentID grade year class0 class1 class2;
  MODEL teff = year01 year12 class0*aclass0 class1*aclass1 class2*aclass2
    / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / TYPE=UN SUBJECT=StudentID;
  REPEATED year / TYPE=VC SUBJECT=StudentID;
  ODS OUTPUT InfoCrit=FixedAcute; RUN;
```

The aclass0-2 effects are the year-specific custom intercepts we created before. The cls0-2 variables are the year-specific class ID variables.

```

Dimensions
Covariance Parameters      2
Columns in X               64
Columns in Z Per Subject  1
Subjects                   486 → number of students
Max Obs Per Subject       3

```

```

Covariance Parameter Estimates
Cov Parm  Subject  Estimate  Standard Error  Z Value  Pr > Z
UN(1,1)   StudentID  0.5960   0.04995        11.93   <.0001
year      StudentID  0.3254   0.01792        18.16   <.0001

```

Class mean differences explained 19.8% of the level-1 residual variance (which caused a 2.7% increase in the level-2 student random intercept variance).

```

Null Model Likelihood Ratio Test
DF  Chi-Square  Pr > ChiSq
1   375.18      <.0001

```

The student random intercept variance is still significantly > 0.

```

Information Criteria
Neg2LogLike  Parms  AIC  AICC  HQIC  BIC  CAIC
2948.9       2    2952.9  2952.9  2956.2  2961.3  2963.3

```

```

Type 3 Tests of Fixed Effects
Effect      Num  Den  F Value  Pr > F
year01      0    .    .         .
year12      0    .    .         .
aclass0*class0  17  884  4.58    <.0001
aclass1*class1  19  881  2.51    0.0004
aclass2*class2  19  890  3.89    <.0001

```

(solution for fixed effects is omitted to save space)

The dots are given because it can't marginalize the year slopes across classes given their year-specific effects.

Although this fixed effects model allows us to control for class mean differences, it does not allow us to predict them. To do that, we instead will use class random intercept variances for each year as follows.

Model 3: Crossed Random Effects for Year-Specific (Acute) Effects of Class (from Equation 11.14)

$$\text{Effort}_{\text{tsc}} = \gamma_{000} + \gamma_{100}(\text{Year01}_{\text{tsc}}) + \gamma_{200}(\text{Year12}_{\text{tsc}}) + U_{0s0} + e_{\text{tsc}} \\ + U_{00c}^0(\text{Int0}_{\text{tsc}}) + U_{00c}^1(\text{Int1}_{\text{tsc}}) + U_{00c}^2(\text{Int2}_{\text{tsc}})$$

```

TITLE1 "Random class effects -- acute";
PROC MIXED DATA=classroom COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
CLASS StudentID grade year class0 class1 class2;
MODEL teff = year01 year12 / SOLUTION DDFM=Satterthwaite OUTPM=PredTime;
RANDOM aclass0 / SUBJECT=class0 TYPE=UN;
RANDOM aclass1 / SUBJECT=class1 TYPE=UN;
RANDOM aclass2 / SUBJECT=class2 TYPE=UN;
RANDOM INTERCEPT / SUBJECT=StudentID TYPE=UN;
REPEATED year / SUBJECT=StudentID TYPE=VC;
ODS OUTPUT InfoCrit=FitPAcute CovParms=CovPAcute; RUN;

```

The aclass0-2 effects in the RANDOM statements are the year-specific custom intercepts we created. The class0-2 are the year-specific class ID variables.

```

Dimensions
Covariance Parameters      5
Columns in X               3
Columns in Z               547
Subjects                   1 → no repetitions of the V matrix (one giant matrix for all cases)
Max Obs Per Subject       1214

```

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
aclass0	class0	0.08082	0.03521	2.30	0.0109	class random intercept variance at year0
aclass1	class1	0.04766	0.02511	1.90	0.0288	class random intercept variance at year1
aclass2	class2	0.08706	0.03782	2.30	0.0107	class random intercept variance at year2
Intercept	StudentID	0.5925	0.04922	12.04	<.0001	student random intercept variance
year	StudentID	0.3262	0.01790	18.22	<.0001	residual within-student, within-class

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
2984.6	5	2994.6	2994.7	2984.6	2984.6	2989.6

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	3.8943	0.06726	25.5	57.90	<.0001
year01	-0.1817	0.09303	29.2	-1.95	0.0605
year12	0.02060	0.09265	30.5	0.22	0.8256

```
%FitTest(FitFewer=FitPInt, FitMore=FitPacute);
Likelihood Ratio Test for FitPInt vs. FitPacute
```

Neg2Log							
Name	Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitPInt	3067.7	2	3071.7	3080.1	.	.	.
FitPacute	2984.6	5	2994.6	2984.6	83.0854	3	0

Model fit improved significantly after adding three level-2 random intercept variances for year-specific class effects (relative to student-only).

Given the lack of covariances among the random effects, we can sum the level-1 residual variance, level-2 student random intercept variance, and the per-year level-2 class random intercept variance to create year-specific total variances with which to compute the proportion of variance due to each source. In doing so, the proportion of variance due to level-1 differences across years was 32.6%, 33.8%, and 32.4% at years 0, 1, and 2, respectively. The proportion of variance due to level-2 student mean differences was 59.3%, 61.3%, and 58.9% at years 0, 1, and 2. Finally, the proportion of variance due to level-2 class mean differences was 8.1%, 4.9%, and 8.7% at years 0, 1, and 2.

A side note: the model above could have also been estimated using this simpler SAS syntax for random classes:

```
CLASS StudentID grade ClassID year;
MODEL teff = year01 year12 / SOLUTION DDFM=Satterthwaite OUTPM=PredTime;
RANDOM INTERCEPT / SUBJECT=ClassID TYPE=UN GROUP=year;
RANDOM INTERCEPT / SUBJECT=StudentID TYPE=UN;
```

The advantage of the three-random-statement syntax is the ability to model separate sets of random effects per year, as well as to model different types of intercept effects, as shown next.

Our previous model hypothesized acute class effects, which were present only when a student is actually in that class. An alternative view is that the impact of a class and its teacher will continue in the future even after a student has left the class. We can test this idea by replacing our year-specific intercepts with “transfer” versions, in which the effect of having been in a year 0 class remains in full during years 1 and 2, and the year 1 effect remains in full during year 2.

Model 4: Crossed Random Effects for Cumulative (Transfer) Effects of Class (same equation)

```
TITLE1 "Random class effects -- transfer";
PROC MIXED DATA=classroom COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
CLASS StudentID grade year class0 class1 class2;
MODEL teff = year01 year12 / SOLUTION DDFM=Satterthwaite;
RANDOM tclass0 / SUBJECT=class0 TYPE=UN;
RANDOM tclass1 / SUBJECT=class1 TYPE=UN;
RANDOM tclass2 / SUBJECT=class2 TYPE=UN;
RANDOM INTERCEPT / SUBJECT=StudentID TYPE=UN;
REPEATED year / SUBJECT=StudentID TYPE=VC;
ODS OUTPUT InfoCrit=FitPTransfer; RUN;
```

The tclass0-2 effects in the RANDOM statements are the cumulative (transfer) custom intercepts we created. The class0-2 variables are the year-specific class ID variables.

Covariance Parameter Estimates						
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	class0	0.03330	0.02443	1.36	0.0864	class random intercept variance at year0
UN(1,1)	class1	0.04040	0.02217	1.82	0.0342	class random intercept variance at year1
UN(1,1)	class2	0.09559	0.04120	2.32	0.0102	class random intercept variance at year0
UN(1,1)	StudentID	0.5573	0.04860	11.47	<.0001	student random intercept variance
year	StudentID	0.3578	0.01943	18.41	<.0001	residual within-student, within-class

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
3023.8	5	3033.8	3033.9	3023.8	3023.8	3028.8

Solution for Fixed Effects						
Effect	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept	3.8427	0.07184	35.6	53.49	<.0001	
year01	-0.1923	0.06187	29.1	-3.11	0.0042	
year12	0.02069	0.08232	21.5	0.25	0.8039	

```
%FitTest(FitFewer=FitPInt, FitMore=FitPTransfer);
```

Likelihood Ratio Test for FitPInt vs. FitPTransfer

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitPInt	3067.7	2	3071.7	3080.1	.	.	.
FitPTransfer	3023.8	5	3033.8	3023.8	43.9253	3	1.5653E-9

Model fit also improved significantly after adding three level-2 random intercept variances for year-specific class effects (relative to student-only).

Given that our **acute versus transfer class effect models** differ in their variance components and are non-nested with the same number of parameters, we can compare them using **AIC and BIC**. **Both were smaller** for the acute model, such that the year-specific class effects were preferred. Accordingly, we can continue by examining the extent to which predictors pertaining to each source of sampling (years, students, and time-varying classes) can explain each pile of variance. Let's fast-forward to the final model that includes effects of class grade, student gender, class gender composition, time-varying teacher-perceived student aggression, student mean aggression, and class mean aggression, each grand-mean-centered (given that there is no single class to serve as a reference for group-mean-centering).

SAS syntax to create predictors:

```
* Get means per class (tag=teacher-perceived student aggression);
PROC SORT DATA=classroom; BY class year; RUN;
PROC MEANS NOPRINT DATA=classroom; BY class year;
    VAR girl tagg; OUTPUT OUT=ClassMeans MEAN(girl tagg)= cmGirl cmTagg; RUN;

* Center class predictors;
DATA classroom; MERGE classroom ClassMeans; BY class year;
    size=_FREQ_; size21=size-21; cmGirl50=cmGirl-.50; cmTagg2=cmTagg-2;
    DROP _FREQ_ _TYPE_; RUN;

* Get TV predictor means per student;
PROC SORT DATA=classroom; BY StudentID year; RUN;
PROC MEANS NOPRINT DATA=classroom; BY StudentID;
    VAR tagg; OUTPUT OUT=KidMeans MEAN(tagg)= smTagg; RUN;

* Center TV and student predictors;
DATA classroom; MERGE classroom KidMeans; BY StudentID;
    smTagg2=smTagg-2; tagg2=tagg-2;
    DROP _FREQ_ _TYPE_; RUN;
```

Model 5: Crossed Random Effects for Students and Classes: Final Conditional Model (from 11.17)

$$\begin{aligned}
 \text{Effort}_{\text{tsc}} = & \gamma_{000} + \gamma_{100}(\text{Year}01_{\text{tsc}}) + \gamma_{200}(\text{Year}12_{\text{tsc}}) + U_{0s0} + e_{\text{tsc}} \\
 & + U_{00c}^0(\text{Int}0_{\text{tsc}}) + U_{00c}^1(\text{Int}1_{\text{tsc}}) + U_{00c}^2(\text{Int}2_{\text{tsc}}) \\
 & + \gamma_{001}^0(\text{G}5\text{v}3_c)(\text{Int}0_{\text{tsc}}) + \gamma_{002}^0(\text{G}5\text{v}4_c)(\text{Int}0_{\text{tsc}}) \\
 & + \gamma_{001}^1(\text{G}6\text{v}4_c)(\text{Int}1_{\text{tsc}}) + \gamma_{002}^1(\text{G}6\text{v}5_c)(\text{Int}1_{\text{tsc}}) \\
 & + \gamma_{001}^2(\text{G}7\text{v}5_c)(\text{Int}2_{\text{tsc}}) + \gamma_{002}^2(\text{G}7\text{v}6_c)(\text{Int}2_{\text{tsc}}) \\
 & + \gamma_{010}(\text{G}irl_s) + \gamma_{003}^0(\overline{\text{G}irl}_c - .50)(\text{Int}0_{\text{tsc}}) \\
 & + \gamma_{003}^1(\overline{\text{G}irl}_c - .50)(\text{Int}1_{\text{tsc}}) + \gamma_{003}^2(\overline{\text{G}irl}_c - .50)(\text{Int}2_{\text{tsc}}) \\
 \text{Time-varying} & \rightarrow + \gamma_{300}(\overline{\text{A}gg}_{sc} - 2) + \gamma_{020}(\overline{\text{A}gg}_s - 2) + \gamma_{004}^0(\overline{\text{A}gg}_c - 2)(\text{Int}0_{\text{tsc}}) \\
 \text{student aggression} & \rightarrow + \gamma_{004}^1(\overline{\text{A}gg}_c - 2)(\text{Int}1_{\text{tsc}}) + \gamma_{004}^2(\overline{\text{A}gg}_c - 2)(\text{Int}2_{\text{tsc}})
 \end{aligned}$$

Class-level differences between grades each year

Student gender and class-level differences in % girls each year

Student mean aggression and class mean aggression each year

```

TITLE1 "Acute random effects + grade + student gender + class gender + TAGG";
PROC MIXED DATA=classroom COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=REML;
  CLASS StudentID grade year class0 class1 class2;
  MODEL teff = year01 year12 grade*aclass0 grade*aclass1 grade*aclass2
    girl cmGirl150*aclass0 cmGirl150*aclass1 cmGirl150*aclass2
    tagg2 smTagg2 cmTagg2*aclass0 cmTagg2*aclass1 cmTagg2*aclass2
    / SOLUTION DDFM=Satterthwaite OUTPM=PredFinal;
  RANDOM aclass0 / SUBJECT=class0 TYPE=UN;
  RANDOM aclass1 / SUBJECT=class1 TYPE=UN;
  RANDOM aclass2 / SUBJECT=class2 TYPE=UN;
  RANDOM INTERCEPT / SUBJECT=StudentID TYPE=UN;
  REPEATED year / SUBJECT=StudentID TYPE=VC;
  ESTIMATE "Y0 G3v4" grade*aclass0 -1 1 0 0 0 grade*aclass1 0 0 0 0 0 grade*aclass2 0 0 0 0 0;
  ESTIMATE "Y0 G3v5" grade*aclass0 -1 0 1 0 0 grade*aclass1 0 0 0 0 0 grade*aclass2 0 0 0 0 0;
  ESTIMATE "Y0 G4v5" grade*aclass0 0 -1 1 0 0 grade*aclass1 0 0 0 0 0 grade*aclass2 0 0 0 0 0;
  ESTIMATE "Y1 G4v5" grade*aclass0 0 0 0 0 0 grade*aclass1 0 -1 1 0 0 grade*aclass2 0 0 0 0 0;
  ESTIMATE "Y1 G4v6" grade*aclass0 0 0 0 0 0 grade*aclass1 0 -1 0 1 0 grade*aclass2 0 0 0 0 0;
  ESTIMATE "Y1 G5v6" grade*aclass0 0 0 0 0 0 grade*aclass1 0 0 -1 1 0 grade*aclass2 0 0 0 0 0;
  ESTIMATE "Y2 G5v6" grade*aclass0 0 0 0 0 0 grade*aclass1 0 0 0 0 0 grade*aclass2 0 0 -1 1 0;
  ESTIMATE "Y2 G5v7" grade*aclass0 0 0 0 0 0 grade*aclass1 0 0 0 0 0 grade*aclass2 0 0 -1 0 1;
  ESTIMATE "Y2 G6v7" grade*aclass0 0 0 0 0 0 grade*aclass1 0 0 0 0 0 grade*aclass2 0 0 0 -1 1;
  ESTIMATE "Between Class Gender at Year 0" girl 1 cmGirl150*aclass0 1;
  ESTIMATE "Between Class Gender at Year 1" girl 1 cmGirl150*aclass1 1;
  ESTIMATE "Between Class Gender at Year 2" girl 1 cmGirl150*aclass2 1;
  ESTIMATE "Between Class TAGG at Year 0" TAGG2 1 cmTAGG2*aclass0 1;
  ESTIMATE "Between Class TAGG at Year 1" TAGG2 1 cmTAGG2*aclass1 1;
  ESTIMATE "Between Class TAGG at Year 2" TAGG2 1 cmTAGG2*aclass2 1;
  ESTIMATE "Between Person TAGG" TAGG2 1 smTAGG2 1;
  ODS OUTPUT InfoCrit=FitFinal CovParms=CovFinal; RUN;

%PseudoR2(NCov=5, CovFewer=CovPAcute, CovMore=CovFinal);
%TotalR2(DV=teff, PredFewer=PredTime, PredMore=PredFinal);

```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	Relative to time-only: (total R2=.360)
aclass0	class0	0.01281	0.01199	1.07	0.1427	→ pseudo-R2 = .841
aclass1	class1	0.02433	0.01601	1.52	0.0642	→ pseudo-R2 = .489
aclass2	class2	0.03659	0.02125	1.72	0.0426	→ pseudo-R2 = .580
Intercept	StudentID	0.3441	0.03080	11.17	<.0001	→ pseudo-R2 = .419
year	StudentID	0.2664	0.01449	18.39	<.0001	→ pseudo-R2 = .183

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
2596.5	5	2606.5	2606.6	2596.5	2596.5	2601.5

Solution for Fixed Effects

Effect	grade	Estimate	Standard Error	DF	t Value	Pr > t	
Intercept		3.5931	0.1159	25.6	31.00	<.0001	
year01		-0.05146	0.1307	26.2	-0.39	0.6970	
year12		0.2437	0.1537	25.1	-1.59	0.1252	
aclass0*grade	3	0.004921	0.1150	20.8	0.04	0.9663	
aclass0*grade	4	0.1181	0.1165	19.1	1.01	0.3231	
aclass0*grade	5	0	
aclass0*grade	6	0	
aclass0*grade	7	0	
aclass1*grade	3	0	
aclass1*grade	4	0.04193	0.1328	22	0.32	0.7551	
aclass1*grade	5	-0.08934	0.1247	22.3	-0.72	0.4810	
aclass1*grade	6	0	
aclass1*grade	7	0	
aclass2*grade	3	0	
aclass2*grade	4	0	
aclass2*grade	5	0.2015	0.1475	19.4	1.37	0.1874	
aclass2*grade	6	0.4110	0.1472	18.4	2.79	0.0119	
aclass2*grad	7	0	
girl		0.07658	0.06301	464	1.22	0.2248	within-class student effect
aclass0*cmGirl150		1.2541	0.7317	11.7	1.71	0.1128	contextual class effect Y0
aclass1*cmGirl150		-0.1972	0.8857	15.7	-0.22	0.8267	contextual class effect Y1
aclass2*cmGirl150		0.01186	0.6591	13.8	0.02	0.9859	contextual class effect Y2
tagg2		-0.6055	0.04390	783	-13.79	<.0001	within-student, within-class effect
smTagg2		-0.1977	0.06243	1058	-3.17	0.0016	contextual student effect
aclass0*cmTagg2		0.004471	0.1087	14.7	0.04	0.9677	contextual class effect Y0
aclass1*cmTagg2		0.1260	0.1372	22.9	0.92	0.3679	contextual class effect Y1
aclass2*cmTagg2		0.06775	0.1808	15.2	0.37	0.7130	contextual class effect Y2

Type 3 Tests of Fixed Effects (redundant ones omitted)

Effect	Num DF	Den DF	F Value	Pr > F
aclass0*grade	2	19.5	0.64	0.5392
aclass1*grade	2	21.3	0.54	0.5911
aclass2*grade	2	18.8	3.92	0.0378

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
Year 0 Grade 3 vs 4	0.1132	0.1188	18.7	0.95	0.3528
Year 0 Grade 3 vs 5	-0.00492	0.1150	20.8	-0.04	0.9663
Year 0 Grade 4 vs 5	-0.1181	0.1165	19.1	-1.01	0.3231
Year 1 Grade 4 vs 5	-0.1313	0.1315	19.8	-1.00	0.3304
Year 1 Grade 4 vs 6	-0.04193	0.1328	22	-0.32	0.7551
Year 1 Grade 5 vs 6	0.08934	0.1247	22.3	0.72	0.4810
Year 2 Grade 5 vs 6	0.2095	0.1399	18.7	1.50	0.1508
Year 2 Grade 5 vs 7	-0.2015	0.1475	19.4	-1.37	0.1874
Year 2 Grade 6 vs 7	-0.4110	0.1472	18.4	-2.79	0.0119
Between Class Gender at Year 0	1.3307	0.7323	11.8	1.82	0.0948
Between Class Gender at Year 1	-0.1206	0.8865	15.7	-0.14	0.8935
Between Class Gender at Year 2	0.08844	0.6601	13.9	0.13	0.8953
Between Class TAGG at Year 0	-0.6011	0.1038	12.1	-5.79	<.0001
Between Class TAGG at Year 1	-0.4795	0.1368	22.3	-3.51	0.0020
Between Class TAGG at Year 2	-0.5378	0.1783	14.3	-3.02	0.0091
Between Person TAGG	-0.8032	0.04846	550	-16.58	<.0001